
Before getting started with dependency injection itself, I want to start with an example of the kinds of problems it is going to be solving.



package MyApp;
use MyFramework;

sub call {
    my $self = shift;

    my $logger = Logger->new(log_file => 'logs/myapp.log');
    $logger->log("connecting to database");

    my $dbh = DBI->connect('dbi:mysql:myapp_db');
    my $hello = $dbh->selectall_arrayref('SELECT * FROM my_table')->[0][0];

    $logger->log("rendering template");
    my $template = Template->new(INCLUDE_PATH => 'root/template');
    $template->process('hello.tt', { hello => $hello }, \(my $output));

    return $output;
}

So you're writing a basic web app - it connects to the database, pulls some information out, and renders it into a template.



package MyApp;
use MyFramework;

has model => (is => 'ro', isa => 'Model', default => sub { Model->new });
has view  => (is => 'ro', isa => 'View',  default => sub { View->new });

sub call {
    my $self = shift;
    my $hello = $self->model->get_hello;
    return $self->view->render($hello);
}

Now you want to start adding new features, but it's hard when everything is hardcoded directly in the code, so you start refactoring. This lets you do things like swap out 
the model object for a mock during testing. Instead of making the logger a global (globals are bad), you move it into a builder in the Model and View classes.



package MyApp;
use MyFramework;

has logger => (
    is => 'ro', isa => 'Logger',
    default => sub { Logger->new }
);

has model => (
    is => 'ro', isa => 'Model', lazy => 1,
    default => sub { Model->new(logger => $_[0]->logger) },
);

has view  => (
    is => 'ro', isa => 'View', lazy => 1,
    default => sub { View->new(logger => $_[0]->logger) },
);

sub call {
    my $self = shift;
    my $hello = $self->model->get_hello;
    return $self->view->render($hello);
}

But that required initializing the logger in multiple places, so you pull the logger initialization out into the main app, and just pass it down into where it's needed.



package MyApp;
use MyFramework;

has dsn     => (is => 'ro', isa => 'Str',    default => 'dbi:mysql:myapp_db');
has tt_root => (is => 'ro', isa => 'Str',    default => 'root/template');
has logger  => (is => 'ro', isa => 'Logger', default => sub { Logger->new });

has model => (
    is => 'ro', isa => 'Model', lazy => 1,
    default => sub {
        my $m = Model->connect($_[0]->dsn);
        $m->set_logger($_[0]->logger);
        return $m;
    },
);

has view  => (
    is => 'ro', isa => 'View', lazy => 1,
    default => sub {
        View->new(logger => $_[0]->logger, tt_root => $_[0]->tt_root);
    },
);

sub call { ... }

But you're still duplicating code in your tests (creating the test model object), so you pull some more things out to allow just specifying a dsn, instead of having to create the 
entire model object. But look at how complicated this is now!



Dependency Injection

So dependency injection allows you to get all of those benefits without it being so much of a mess. You just describe the relationships between your classes, and the 
framework handles all of the actual object construction.



Dependency Injection

‣ a form of inversion of control

7

7
So dependency injection allows you to get all of those benefits without it being so much of a mess. You just describe the relationships between your classes, and the 
framework handles all of the actual object construction.



Dependency Injection

‣ a form of inversion of control
‣ "the inverse of garbage collection"

7

7
So dependency injection allows you to get all of those benefits without it being so much of a mess. You just describe the relationships between your classes, and the 

So dependency injection allows you to get all of those benefits without it being so much of a mess. You just describe the relationships between your classes, and the 
framework handles all of the actual object construction.

This makes things like testing much easier, because you can just ask the framework for a model object, and it will create one in the exact same way that it does when your 
application asks for one.

This makes things like testing much easier, because you can just ask the framework for a model object, and it will create one in the exact same way that it does when your 
application asks for one.

Catalyst contains just enough of a dependency injection system to handle its specific application structure, and nothing more. This typically works fine, except when you 
want to do something outside of that structure (reuse a model class in a helper script, for instance).

Catalyst
‣ it is, however, pretty limited

‣ can only create objects, and these objects 
must be either models or views (typically with 
specific class names)

‣ all data to create the objects must be 
specified in the configuration, objects can't 
take other objects as constructor parameters

Catalyst contains just enough of a dependency injection system to handle its specific application structure, and nothing more. This typically works fine, except when you 
want to do something outside of that structure (reuse a model class in a helper script, for instance).

Bread::Board

Bread::Board is a fully-featured, general purpose dependency injection system. It provides a common set of functionality so that applications can be built in a consistent 
way, and interoperate cleanly.



my $c = container MyApp => as {
    service dsn    => 'dbi:mysql:myapp_db';
    service logger => (class => 'Logger', lifecycle    => 'Singleton');
    service view   => (class => 'View',   dependencies => ['logger']);

    service model => (
        class        => 'Model',
        dependencies => ['logger', 'dsn'],
        block        => sub {
            my $m = Model->connect($_[0]->param('dsn'));
            $m->set_logger($_[0]->param('logger'));
            return $m;
        },
    );

    service app => (
        class        => 'MyApp',
        dependencies => ['model', 'view'],
    );
};

$c->resolve(service => 'app');

This is an example of a Bread::Board container. It will be explained in the following slides.



Services

The services are the representation of the data that you want to create.

The services are the representation of the data that you want to create.

service view => (
    class => 'View',
);

14
This is a service which will create a new instance of the class View for you.



Bread::Board::BlockInjection

service model => (
    class => 'Model', # optional
    block => sub {
        my $m = Model->new
        $m->initialize;
        return $m;
    },
);

15
This is a service which will run the given block to create the object.



Bread::Board::Literal

service dsn => 'dbi:mysql:myapp_db';

16
This is a service which just returns the literal data given every time.

Containers are used to organize groups of services. Services within a given container can be used to help build each other.

dependencies.

Services and containers within a container are referred to via service paths, which should be a familiar metaphor. This is used when calling ->fetch or ->resolve, and is also used when specifying 
dependencies.

Dependencies are how you tell the services how to help build each other. Any objects that are necessary to build a given object must be specified as dependencies on a service which provides 
the required object.

Dependencies are how you tell the services how to help build each other. Any objects that are necessary to build a given object must be specified as dependencies on a service which provides 
the required object.



Dependencies

service logger => (class => 'Logger');
service view => (
    class        => 'View',
    dependencies => ['logger'],
);

20
In this example, a logger object will be created via Logger->new, and then the result of that will be passed as View->new(logger => $logger). Note the shortcut for the dependency specification - 
['logger'] is equivalent to { logger => 'logger' }.



Dependencies

service dsn => 'dbi:mysql:myapp_db';
service model => (
    class        => 'Model',
    dependencies => ['dsn'],
    block        => sub {
        my $service = shift;
        return Model->connect($service->param('dsn'));
    },
);

21
Here, a dependency is specified for a block injector service. In this case, the service itself is passed as an argument to the block, after having all of its dependencies resolved. The resolved 
dependencies can be accessed via the 'param' method.



Dependencies

container MyApp => as {
    container Model => as {
        service dsn => 'dbi:mysql:myapp_db';
        service model => (
            class        => 'Model',
            dependencies => ['dsn'],
            block        => sub {
                my $service = shift;
                return Model->connect($service->param('dsn'));
            },
        );
    };
    service app => (
        class        => 'MyApp',
        dependencies => ['Model/model'],
    );
};

22
In this more complicated example, we have the model being held in a subcontainer, which means that the path to the service has to include components for both the container ('Model') and 
service ('model'). The dependency specification of ['Model/model'] is equivalent to { model => 'Model/model' }.

In particular, any place where I previously talked about dependencies in a context other than how to supply them (being passed to constructors, being accessed via the ->param method, etc) can 
also use parameters.

In particular, any place where I previously talked about dependencies in a context other than how to supply them (being passed to constructors, being accessed via the ->param method, etc) can 

my $c = container MyApp => as {
    service user => (
        class      => 'User',
        parameters => {
            name => { isa => 'Str' },
        },
    );
};

$c->resolve(
    service    => 'user', 
    parameters => { name => 'doy' }
);

24
Parameters are specified with a map of parameter names to parameter options. Valid options are 'isa', 'default', and 'optional'.



Parameters

my $c = container MyApp => as {
    service user => (
        class      => 'User',
        parameters => {
            name => { isa => 'Str' },
        },
    );
    service superusers => (
        block        => sub { [ $_[0]->param('root') ] },
        dependencies => {
            root => { user => { name => 'root' } },
        },
    );
};

25
Here, a service has a dependency on a service with a required parameter. The parameter can also be supplied as part of the dependency specification.

my $c = container MyApp => as {
    service user => (
        class => 'User',
        parameters => {
            name => { isa => 'Str' },
        },
    );
    service superusers => (
        block => sub {
            [ $_[0]->param('user')->inflate(name => 'root') ]
        },
        dependencies => ['user'],
    );
};

26
Often though, the parameter value isn't known at the time that the container is being constructed. If a dependency on a parameterized service is specified without the parameter being provided, 
then the result of resolving the service is a special object, whose purpose is only to accept the parameters (via the 'inflate' method) and return the actual resolved value from the service.



Parameters

my $c = container MyApp => as {
    service user => (
        class      => 'User',
        parameters => {
            name => { 
                isa      => 'Str', 
                optional => 1, 
                default  => 'guest' 
            },
        },
    );
};
# user with name 'guest'
$c->resolve(service => 'user');

# user with name 'doy'
$c->resolve(service => 'user', parameters => { name => 'doy' });

27
Parameters can also be given defaults. In this case, you typically also want to make them optional.



Parameters

my $c = container MyApp => as {
    service default_username => 'guest';
    service user => (
        class => 'User',
        parameters => {
            name => { isa => 'Str', optional => 1 },
        },
        dependencies => {
            name => 'default_username',
        },
    );
};
# user with name 'guest'
$c->resolve(service => 'user');

# user with name 'doy'
$c->resolve(service => 'user', parameters => { name => 'doy' });

28
Parameters and dependencies can both be specified for a service. If an optional parameter is given with the same name as a dependency, it can be used to provide a more powerful mechanism 
for specifying a default. The parameter will be used if provided when the service is resolved, otherwise the dependency will be resolved as normal and used.

29
Lifecycles affect how the service is resolved, on a per-service basis. Creating singleton services rather than singleton classes is much better for reuse.

Lifecycles

when creating the service, the same object will 
be returned each time

Lifecycles affect how the service is resolved, on a per-service basis. Creating singleton services rather than singleton classes is much better for reuse.



my $c = container MyApp => as {
    service dsn    => 'dbi:mysql:myapp_db';
    service logger => (class => 'Logger', lifecycle    => 'Singleton');
    service view   => (class => 'View',   dependencies => ['logger']);

    service model => (
        class        => 'Model',
        dependencies => ['logger', 'dsn'],
        block        => sub {
            my $m = Model->connect($_[0]->param('dsn'));
            $m->set_logger($_[0]->param('logger'));
            return $m;
        },
    );

    service app => (
        class        => 'MyApp',
        dependencies => ['model', 'view'],
    );
};

$c->resolve(service => 'app');

30

So now this example from earlier should make more sense.

Having to manually specify dependencies for every service does get tedious after a while as well. You're already listing these things once when writing your classes (via attributes and type 
constraints), so it seems a bit redundant to also have to do it again in the Bread::Board configuration. This is what typemaps solve.

‣ defines a mapping from a class_type to a 

service

can request an object of a particular type:     
$c->resolve(type => 'Model');

‣ with this, we can (mostly) infer the 
dependencies for a given class

Having to manually specify dependencies for every service does get tedious after a while as well. You're already listing these things once when writing your classes (via attributes and type 
constraints), so it seems a bit redundant to also have to do it again in the Bread::Board configuration. This is what typemaps solve.

package Model
use Moose;
has logger => (is => 'ro', isa => 'Logger', required => 1);

package Logger;
use Moose;

my $c = container MyApp => as {
    typemap Logger => infer;
    typemap Model  => infer;
};

$c->resolve(type => 'Model')->logger; # a valid logger object

32
Here, we just tell Bread::Board to 'infer' the dependencies for a given service. It automatically creates a constructor injection service for the specified class, and populates the dependencies by 
introspecting the class.

Generally when writing your classes, you'll want to make as many things required as possible, moving all of the logic involved in constructing each of the attributes outside of the class. If you do 
this, inference works very smoothly.

this, inference works very smoothly.



Inferred Services
‣ required attributes are automatically inferred, becoming 

parameters, or specified manually as dependencies

Generally when writing your classes, you'll want to make as many things required as possible, moving all of the logic involved in constructing each of the attributes outside of the class. If you do 
this, inference works very smoothly.



my $c = container MyApp => as {
    service dsn    => 'dbi:mysql:myapp_db';
    typemap Logger => infer(lifecycle => 'Singleton');
    typemap View   => infer;

    service model => (
        class        => 'Model',
        dependencies => ['type:Logger', 'dsn'],
        block        => sub {
            my $m = Model->connect($_[0]->param('dsn'));
            $m->set_logger($_[0]->param('type:Logger'));
            return $m;
        },
    );

    typemap Model => 'model';

    typemap MyApp => infer;
};

$c->resolve(type => 'MyApp');

34

Using typemaps, the previous example could be rewritten like this. Note that typemaps can be specified for non-inferred services too. Manual dependencies on types are 
indicated by the 'type:' prefix.



Catalyst and Bread::Board

35

package MyApp;
use Catalyst 'Bread::Board';

__PACKAGE__->config(
    'Plugin::Bread::Board' => {
        container => MyApp::Container->new,
    }
);

35
As mentioned earlier, Catalyst uses its own dependency injection system, which isn't very featureful. Catalyst::Plugin::Bread::Board lets you replace that with Bread::Board. There is also currently 
work being done to merge this into Catalyst itself.



Best Practices

36

Containers are for initialization

36
A container that you can access at runtime is effectively the service locator pattern, rather than dependency injection. This loses the benefit of separating out which parts of your code need 
which other parts, since you can get any service you want out of the container. Also, factories aren't as bad as they sometimes have to be in other languages - closures make perfectly good 
factories for simple cases.

Containers are for initialization

36

Containers are for initialization
package MyApp::Container;
use Moose;
extends 'Bread::Board::Container';

sub BUILD {
    container $self => as {
        # ...
    };
}

37
The container sugar not only works to create new containers, but also to append to existing containers. Just pass a container instance as the first parameter instead of a name.

Containers are for initialization

container SomethingElse => as {
    container MyApp::Container->new;
};

38
Putting this configuration in BUILD means that you can create new fully-configured container instances by calling ->new.

package MyApp::Container;
use Moose;
use Bread::Board::Declare;

has dsn    => (is => 'ro', isa => 'Str', value => 'dbi:mysql:myapp_db');
has logger => (is => 'ro', isa => 'Logger');
has view   => (is => 'ro', isa => 'View', infer => 1);

has model => (
    is           => 'ro',
    isa          => 'Model',
    infer        => 1,
    dependencies => ['dsn'],
    block        => sub {
        my $m = Model->connect($_[0]->param('dsn'));
        $m->set_logger($_[0]->param('logger'));
        return $m;
    },
);

has app => (is => 'ro', isa => 'MyApp', infer => 1);

39
If you need to create more complicated container classes, doing everything in BUILD may still be uglier than necessary. Bread::Board::Declare provides a nicer sugar layer for doing that, by 
creating services from attributes. This is what the original example would look like in Bread::Board::Declare.



Bread::Board::Declare

40

40
This solves the problem of having to modify the container definition itself whenever you want to swap out individual services, which tends to be something that comes up a lot as you use a 
container in more than one place (including during testing).

package MyApp;
use OX;

has model => (is => 'ro', isa => 'Model');
has view  => (is => 'ro', isa => 'View');

has controller => (
    is    => 'ro',
    isa   => 'Controller',
    infer => 1,
);

router as {
    route '/' => 'controller.index';
};

42
OX is a new web framework that is built around Bread::Board, that uses Bread::Board::Declare as a sugar layer and Path::Router for dispatching. In this example, a request for '/' would cause a 
Controller object to be resolved (probably with the help of new Model and View objects), and then the 'index' method would be called on that object to handle the request.

# equivalent to MyApp->new->resolve(service => 'App');
my $psgi_app = MyApp->new->to_app;

# attributes work the same way they do in Bread::Board::Declare
my $model = MyApp->new->model;

43
The app itself is just a Bread::Board container, so it can be used as a subcontainer in other apps, and resolving services works as expected. Bread::Board::Declare handles resolving attributes, so 
reusing parts of the container in tests or external scripts becomes trivial.

package MyApp;
use OX;

has logger => (is => 'ro', isa => 'Logger');

has model => (is => 'ro', isa => 'Model', infer => 1);
has view  => (is => 'ro', isa => 'View',  infer => 1);

has controller => (
    is    => 'ro',
    isa   => 'Controller',
    infer => 1,
);

router as {
    route '/' => 'controller.index';
};

44
The Bread::Board framework that OX is based on can be used to create objects that have nothing to do with the typical Model/View/Controller layout - in fact, the names used by these attributes 
are just convention, there is nothing special going on here at all.

https://metacpan.org/module/Bread::Board

https://github.com/stevan/OX

